```
> restart:
```

Ranada2.mws (RMKiehn 9/10/97)

A Propagating solution with Longitudinal B. By modifying the Ranada stereographic example slightly it is possible to generate a propagating solution to the Maxwell Faraday equations that is irreducibly 3 dimensional. The Electric field is always transverse to the surface of propagating phase, but the B field is not.

```
>
>  restart;with(liesymm):with(linalg):with(plots):setup(x,y,z,t);
Warning, new definition for close
Warning, new definition for norm
Warning, new definition for trace
```

$$[x, y, z, t]$$

```
> defform(x=0,y=0,z=0,t=0,a=const,b=const,c=const,k=const,mu=const,omega=const,m=const);
```

$$\mathrm{defform}(x=0, y=0, z=0, t=0, a=const, b=const, c=const, k=const, \mu=const, \omega=const, m=const)$$

```
> dR:=[d(x),d(y),d(z),d(t)];
```

$$dR := [\,\mathrm{d}(x), \mathrm{d}(y), \mathrm{d}(z), \mathrm{d}(t)\,]$$

Define a characteristic denominator, and then specify the four functions that are the covariant components of the Action 1-form.

```
> p:=2;e:=0;c:=1;a:=1;b:=1;n:=4;rrxy:=1+a*(x)^p+b*(y)^p+c*(z)^p+e*t^p:ff:=cos(k*z-omega*t)/rrxy^(n/p);
> ;
```

$$p := 2$$

$$e := 0$$

$$c := 1$$

$$a := 1$$

$$b := 1$$

$$n := 4$$

$$ff := \frac{\cos(-k\,z+\omega\,t)}{(1+x^2+y^2+z^2)^2}$$

A choice for the 4 potentials. Note that p has been chosen as 2 to yield the quadratic form, n is taken to be 4 to fit the Ranada format, the phase propagation is presumed to be in the positive z direction, and the speed of propagation is omega/k.

```
> A1:=y*ff;A2:=-x*ff;A3:=k*cos(k*z-omega*t);A4:=omega*cos(k*z-omega*t);
```

$$A1 := \frac{y\cos(-k\,z+\omega\,t)}{(1+x^2+y^2+z^2)^2}$$

$$A2 := -\frac{x\cos(-k\,z+\omega\,t)}{(1+x^2+y^2+z^2)^2}$$

$$A3 := k\cos(-k\,z+\omega\,t)$$

$$A4 := \omega\cos(-k\,z+\omega\,t)$$

```
> Action:=A1*d(x)+A2*d(y)+A3*d(z)-A4*d(t);
```

$$Action := \frac{y\,\%1\,\mathrm{d}(x)}{(1+x^2+y^2+z^2)^2} - \frac{x\,\%1\,\mathrm{d}(y)}{(1+x^2+y^2+z^2)^2} + k\,\%1\,\mathrm{d}(z) - \omega\,\%1\,\mathrm{d}(t)$$

$$\%1 := \cos(-k\,z+\omega\,t)$$

```
> F:=wcollect(d(Action));
```

$$F := \frac{y\,(-k\,\%1 - k\,\%1\,x^2 - k\,\%1\,y^2 - k\,\%1\,z^2 + 4\,\%2\,z)\,((\mathrm{d}(x))\,\&\hat{}\,(\mathrm{d}(z)))}{(1+x^2+y^2+z^2)^3}$$
$$+ \left(-\frac{\%2\,(1-3\,x^2+y^2+z^2)}{(1+x^2+y^2+z^2)^3} - \frac{\%2\,(1+x^2-3\,y^2+z^2)}{(1+x^2+y^2+z^2)^3}\right)((\mathrm{d}(x))\,\&\hat{}\,(\mathrm{d}(y)))$$
$$- \frac{x\,(-k\,\%1 - k\,\%1\,x^2 - k\,\%1\,y^2 - k\,\%1\,z^2 + 4\,\%2\,z)\,((\mathrm{d}(y))\,\&\hat{}\,(\mathrm{d}(z)))}{(1+x^2+y^2+z^2)^3} - \frac{x\,\%1\,\omega\,((\mathrm{d}(y))\,\&\hat{}\,(\mathrm{d}(t)))}{(1+x^2+y^2+z^2)^2}$$
$$+ \frac{y\,\%1\,\omega\,((\mathrm{d}(x))\,\&\hat{}\,(\mathrm{d}(t)))}{(1+x^2+y^2+z^2)^2}$$
$$\%1 := \sin(-k\,z + \omega\,t)$$
$$\%2 := \cos(-k\,z + \omega\,t)$$

F is the electromagnetic 2-form in covariant language for all diffeomorphisms. It is gauge invariant with respect to all closed 1-form additions to the 1-form of Action.

The three components of the Vector potential are:

```
> A:=[A1,A2,A3];
```

$$A := \left[\frac{y\cos(-k\,z+\omega\,t)}{(1+x^2+y^2+z^2)^2}, -\frac{x\cos(-k\,z+\omega\,t)}{(1+x^2+y^2+z^2)^2}, k\cos(-k\,z+\omega\,t)\right]$$

The three components of the Magnetic field are:

```
> B:=(curl(A,[x,y,z])):B1:=(B[1]);B2:=(B[2]);B3:=(B[3]);
```

$$B1 := \frac{x\sin(-k\,z+\omega\,t)\,k}{(1+x^2+y^2+z^2)^2} - 4\,\frac{x\cos(-k\,z+\omega\,t)\,z}{(1+x^2+y^2+z^2)^3}$$
$$B2 := \frac{y\sin(-k\,z+\omega\,t)\,k}{(1+x^2+y^2+z^2)^2} - 4\,\frac{y\cos(-k\,z+\omega\,t)\,z}{(1+x^2+y^2+z^2)^3}$$
$$B3 := -2\,\frac{\cos(-k\,z+\omega\,t)}{(1+x^2+y^2+z^2)^2} + 4\,\frac{x^2\cos(-k\,z+\omega\,t)}{(1+x^2+y^2+z^2)^3} + 4\,\frac{y^2\cos(-k\,z+\omega\,t)}{(1+x^2+y^2+z^2)^3}$$

Note that B3 is not zero and is perpendicular to the propagating phase front.

The three components of the Electric Field are:

```
>
> E:=[-diff(A4,x)-diff(A[1],t),-diff(A4,y)-diff(A[2],t),-diff(A4,z)-diff(A[3],t)]:
  E1:=factor(E[1]);E2:=factor(E[2]);E3:=factor(E[3]);
```

$$E1 := \frac{y\sin(-k\,z+\omega\,t)\,\omega}{(1+x^2+y^2+z^2)^2}$$
$$E2 := -\frac{x\sin(-k\,z+\omega\,t)\,\omega}{(1+x^2+y^2+z^2)^2}$$
$$E3 := 0$$

It would appear that this "wave" is TE (E in the direction of advancing phase is zero.)

The Topological Parity 4 form (Second Poncare invariant) vanishes implying that the 1 form of Action defines a contact manifold. Therefore the integral of the 3-form of topological torsion over a closed domain, not a boundary, is a topological invariant.

```
> EdotB:=factor(innerprod(E,B));EdotE:=innerprod(E,E);
```

$$EdotB := 0$$

$$EdotE := \frac{\sin(-k\,z + \omega\,t)^2\,\omega^2\,(y^2 + x^2)}{(1 + x^2 + y^2 + z^2)^4}$$

The Torsion current.

```
> ExA:=crossprod(E,A):Bphi:=[B1*A4,B2*A4,B3*A4]:
> TORS:=evalm(ExA+A4*B):
> AdotB:=factor(inner(A,B));
```

$$AdotB := -2\,\frac{k\cos(-k\,z + \omega\,t)^2\,(1 - x^2 - y^2 + z^2)}{(1 + x^2 + y^2 + z^2)^3}$$

```
> TORSION:=[factor(TORS[1]),factor(TORS[2]),factor(TORS[3]),AdotB];
```

$$TORSION := \left[-4\,\frac{x\,\omega\,\%1^2\,z}{(1 + x^2 + y^2 + z^2)^3}, -4\,\frac{\omega\,y\,\%1^2\,z}{(1 + x^2 + y^2 + z^2)^3}, -2\,\frac{\%1^2\,(1 - x^2 - y^2 + z^2)\,\omega}{(1 + x^2 + y^2 + z^2)^3}, -2\,\frac{k\,\%1^2\,(1 - x^2 - y^2 + z^2)}{(1 + x^2 + y^2 + z^2)^3}\right]$$

$$\%1 := \cos(-k\,z + \omega\,t)$$

The 4 vector of Torsion Current (and the fourth component which is the helicity) are proportional to the SQUARE of the oscillatory term phase term.

The 4Divergence of the Torsion current is:

```
> DIV4DofT:=factor(diverge(TORSION,[x,y,z,t]));
```

$$DIV4DofT := 0$$

establishing the "conservation law" for topological torsion.

Note that none of this assumed a group constraint, nor an assumption of "duality" to produce an E field from a B field (as used by Ranada).

These fields satisfy the Maxwell-Faraday equations for E and B, by construction!

No mention is made of D and H, nor of the charge distributions and constituive relations that might be compatable with these field intensities. However, as the Torsion Current is closed, it is a candidate for the charge current distribution of classical electromagnetic theory. The problem is to find solutions to the partial differential equations, curlH-partialdD/dt = T and divD = A.B, where the RHS is given. Another day.

RMK.